

F01ZSFP

NAG Parallel Library Routine Document

Note: Before using this routine, please read the Users' Note for your implementation to check for implementation-dependent details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

F01ZSFP distributes an m by n real matrix A on a logical grid of processors in a cyclic 2-d block distribution.

This routine distributes matrices in the form required by a number of the F04 Black Box routines. A user-supplied subroutine is required to generate a block of the matrix A .

2 Specification

```

SUBROUTINE F01ZSFP(ICNTXT, GMAT, M, N, NB, A, LDA, IFAIL)
DOUBLE PRECISION  A(LDA, *)
INTEGER           ICNTXT, M, N, NB, LDA, IFAIL
EXTERNAL         GMAT

```

3 Data Distribution

3.1 Definitions

The following definitions are used in describing the data distribution within this document:

- m_p – the number of rows in the logical processor grid.
- n_p – the number of columns in the logical processor grid.
- p_r – the row grid coordinate of the calling processor.
- p_c – the column grid coordinate of the calling processor.
- N_b – the blocking factor for the distribution of the rows and columns of the matrix.
- $\text{numroc}(\alpha, b_\ell, q, s, k)$ – a function which gives the **number of rows or columns** of a distributed matrix owned by the processor with the row or column coordinate q (p_r or p_c), where α is the total number of rows or columns of the matrix, b_ℓ is the blocking factor used (N_b), s is the row or column coordinate of the processor that possesses the first row or column of the distributed matrix and k is either m_p or n_p . The Library provides the function Z01CAFP for the evaluation of numroc.

3.2 Global and Local Arguments

The input arguments M, N, NB and IFAIL are global and so must have the same value on entry to the routine on each processor. The output argument IFAIL is global and so will return the same value on exit from the routine on each processor. The external procedure GMAT is global. The remaining arguments are local.

4 Arguments

- 1: ICNTXT — INTEGER *Local Input*
On entry: the BLACS context used by the communication mechanism, usually returned by a call to Z01AAFP.
- 2: GMAT — SUBROUTINE, supplied by the user. *Global External Procedure*
 GMAT must return the block $A(i_1 : i_2, j_1 : j_2)$ of the matrix to be distributed, in the array AL.

Its specification is:

SUBROUTINE	GMAT(I1, I2, J1, J2, AL, LDAL)	
DOUBLE PRECISION	AL(LDAL,*)	
INTEGER	I1, I2, J1, J2, LDAL	
1:	I1 — INTEGER	<i>Local Input</i>
	<i>On entry:</i> i_1 , the first row of the block of A to be generated.	
2:	I2 — INTEGER	<i>Local Input</i>
	<i>On entry:</i> i_2 , the last row of the block of A to be generated.	
3:	J1 — INTEGER	<i>Local Input</i>
	<i>On entry:</i> j_1 , the first column of the block of A to be generated.	
4:	J2 — INTEGER	<i>Local Input</i>
	<i>On entry:</i> j_2 , the last column of the block of A to be generated.	
5:	AL(LDAL,*) — DOUBLE PRECISION array	<i>Local Output</i>
	<i>On exit:</i> AL must contain the block $A(i_1 : i_2, j_1 : j_2)$ of the matrix A in its first $(I2-I1+1)$ rows and $(J2-J1+1)$ columns.	
6:	LDAL — INTEGER	<i>Local Input</i>
	<i>On entry:</i> the first dimension of the array AL.	

GMAT must be declared as EXTERNAL in the (sub)program from which F01ZSFP is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 3:** M — INTEGER *Global Input*
On entry: the number of rows of the matrix A , m .
Constraint: $M \geq 0$.
- 4:** N — INTEGER *Global Input*
On entry: the number of columns of the matrix A , n .
Constraint: $N \geq 0$.
- 5:** NB — INTEGER *Global Input*
On entry: the blocking factor for distributing the matrix A , N_b .
Constraint: $NB \geq 1$.
- 6:** A(LDA,*) — DOUBLE PRECISION array *Local Output*
Note: the second dimension of the array A must be at least $\max(1, \text{numroc}(N, NB, p_c, 0, n_p))$.
On exit: the local part of the matrix A , distributed in a cyclic 2-d block fashion.
- 7:** LDA — INTEGER *Local Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F01ZSFP is called.
Constraint: $LDA \geq \max(1, \text{numroc}(M, NB, p_r, 0, m_p))$

8: IFAIL — INTEGER*Global Input/Global Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in the Essential Introduction) the recommended values are:

IFAIL = 0, if multigridding is **not** employed;
 IFAIL = -1, if multigridding is employed.

On exit: IFAIL = 0 unless the routine detects an error (see Section 5).

5 Errors and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output from the root processor (or processor {0,0} when the root processor is not available) on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = -2000

The routine has been called with an invalid value of ICNTXT on one or more processors.

IFAIL = -1000

The logical processor grid and library mechanism (Library Grid) have not been correctly defined, see Z01AAFP.

IFAIL = -*i*

On entry, the *i*th argument was invalid. This error occurred either because a global argument did not have the same value on all logical processors, or because its value on one or more processors was incorrect. An explanatory message distinguishes between these two cases.

6 Further Comments

This routine may be used to distribute the data in the form required by a number of the Black Box routines in Chapter F04. This routine may also be used to distribute a vector.

6.1 Algorithmic Detail

The routine generates the matrix on a logical processor by a cyclic 2-d block distribution.

6.2 Parallelism Detail

The routine generates the matrix on each logical processor independently.

7 References

- [1] Dongarra J J and Whaley R C (1995) A users' guide to the BLACS v1.0. *LAPACK Working Note 94 (Technical Report CS-95-281)* Department of Computer Science, University of Tennessee, 107 Ayres Hall, Knoxville, TN 37996-1301, USA.
 URL: <http://www.netlib.org/lapack/lawns/lawn94.ps>

8 Example

To solve the equations

$$Ax = b,$$

where the matrix *A* is given by

$$a_{ij} = \min(i, j)$$

and the right-hand side is the vector

$$b = \begin{pmatrix} 4 \\ 7 \\ 9 \\ 10 \end{pmatrix}$$

The exact solution is the vector $(1, 1, 1, 1)^T$. The Black Box routine F04FBFP is called to solve the equations following the distribution of A by a call to F01ZSFP and a call to X04BGFP to read b from a data file. The routine X04BHFP is used to print the solution vector x . The example illustrates the solution for four equations, with a 2 by 2 logical grid of processors and a block size of 2.

Note: the listing of the Example Program presented below does not give a full pathname for the data file being opened, but in general the user must give the full pathname in this and any other OPEN statement.

8.1 Example Text

```
*      F01ZSFP Example Program Text
*      NAG Parallel Library Release 2. NAG Copyright 1996.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX
PARAMETER       (NMAX=100)
INTEGER          MG, NG
PARAMETER       (MG=2,NG=2)
INTEGER          NB
PARAMETER       (NB=2)
INTEGER          LDA, TDA, TDB
PARAMETER       (LDA=NMAX/MG+NB,TDA=NMAX/NG+NB,TDB=1)
INTEGER          LDB
PARAMETER       (LDB=LDA)
*      .. Local Scalars ..
INTEGER          ICNTXT, IFAIL, N, NCOLS, NROWS
LOGICAL         ROOT
*      .. Local Arrays ..
DOUBLE PRECISION A(LDA,TDA), B(LDB,TDB), WORK(NMAX)
*      .. External Functions ..
LOGICAL         Z01ACFP
EXTERNAL        Z01ACFP
*      .. External Subroutines ..
EXTERNAL        F01ZSFP, F04FBFP, GMATA, X04BGFP, X04BHFP,
+              Z01AAFP, Z01ABFP
*      .. Executable Statements ..
ROOT = Z01ACFP()
IF (ROOT) WRITE (NOUT,*) 'F01ZSFP Example Program Results'
*
NROWS = MG
NCOLS = NG
IFAIL = 0
*
CALL Z01AAFP(ICNTXT,NROWS,NCOLS,IFAIL)
*
OPEN (NIN,FILE='f01zsfpe.d')
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
*
*      Generate the matrix A
*
```

```

        IFAIL = 0
        CALL F01ZSFP(ICNTXT,GMATA,N,N,NB,A,LDA,IFAIL)
*
*   Read in the right hand side
*
        IFAIL = 0
        CALL X04BGFP(ICNTXT,NIN,N,1,NB,B,LDB,IFAIL)
*
*   Solve the system
*
        IFAIL = 0
        CALL F04FBFP(ICNTXT,N,NB,A,LDA,1,B,LDB,IFAIL)
*
*   Print the results
*
        IFAIL = 0
        CALL X04BHFP(ICNTXT,NOUT,N,1,NB,B,LDB,'(1X,F7.3)',WORK,IFAIL)
*
        IFAIL = 0
        CALL Z01ABFP(ICNTXT,'N',IFAIL)
*
        STOP
        END
*
        SUBROUTINE GMATA(I1,I2,J1,J2,AL,LDAL)
*   GMATA generates the block A(I1: I2, J1: J2) of the matrix A such
*   that
*
*       a(i,j) = min(i,j)
*
*   in the array AL.
*
*   .. Scalar Arguments ..
        INTEGER          I1, I2, J1, J2, LDAL
*   .. Array Arguments ..
        DOUBLE PRECISION AL(LDAL,*)
*   .. Local Scalars ..
        INTEGER          I, J, K, L
*   .. Intrinsic Functions ..
        INTRINSIC        MIN
*   .. Executable Statements ..
        L = 1
        DO 40 J = J1, J2
            K = 1
            DO 20 I = I1, I2
                AL(K,L) = MIN(I,J)
                K = K + 1
            20    CONTINUE
                L = L + 1
        40    CONTINUE
*
        RETURN
        END

```

8.2 Example Data

```

F01ZSFP Example Program Data
4      : Value of N

```

```
4.0  
7.0  
9.0  
10.0 : End of right hand side vector B
```

8.3 Example Results

```
F01ZSFP Example Program Results  
1.000  
1.000  
1.000  
1.000
```
